<commit_before>{ lib, fetchFromGitHub, python3Packages, gettext, gdk-pixbuf
, gobject-introspection, gtk3, wrapGAppsHook }:

with lib;

python3Packages.buildPythonApplication rec {
  pname = "nicotine-plus";
  version = "3.1.1";

  src = fetchFromGitHub {
    owner = "Nicotine-Plus";
    repo = "nicotine-plus";
    rev = version;
    hash = "sha256-NfI2RfxAYhA1qefml1ayfYWjbkrzUL4l9p2Rm/ROnzQ=";
  };

  nativeBuildInputs = [ gettext wrapGAppsHook ];

  propagatedBuildInputs = [ gtk3 gdk-pixbuf gobject-introspection ]
    ++ (with python3Packages; [ pygobject3 ]);

  postInstall = ''
    mv $out/bin/nicotine $out/bin/nicotine-plus
  '';

  doCheck = false;

  meta = {
    description = "A graphical client for the SoulSeek peer-to-peer system";
    homepage = "https://www.nicotine-plus.org";
    license = licenses.gpl3Plus;
    maintainers = with maintainers; [ ehmry klntsky ];
    platforms = platforms.unix;
  };
}
<commit_msg>nicotine-plus: Fix crash when opening file dialog, closes #142014<commit_after>{ lib, fetchFromGitHub, python3Packages, gettext, gdk-pixbuf
, gobject-introspection, gtk3, wrapGAppsHook }:

with lib;

python3Packages.buildPythonApplication rec {
  pname = "nicotine-plus";
  version = "3.1.1";

  src = fetchFromGitHub {
    owner = "Nicotine-Plus";
    repo = "nicotine-plus";
    rev = version;
    hash = "sha256-NfI2RfxAYhA1qefml1ayfYWjbkrzUL4l9p2Rm/ROnzQ=";
  };

  nativeBuildInputs = [ gettext wrapGAppsHook ];

  propagatedBuildInputs = [ gtk3 gdk-pixbuf gobject-introspection ]
    ++ (with python3Packages; [ pygobject3 ]);

  postInstall = ''
    mv $out/bin/nicotine $out/bin/nicotine-plus
  '';

  preFixup = ''
    gappsWrapperArgs+=(
      --prefix XDG_DATA_DIRS : "${gtk3}/share/gsettings-schemas/${gtk3.name}"
    )
  '';

  doCheck = false;

  meta = {
    description = "A graphical client for the SoulSeek peer-to-peer system";
    homepage = "https://www.nicotine-plus.org";
    license = licenses.gpl3Plus;
    maintainers = with maintainers; [ ehmry klntsky ];
    platforms = platforms.unix;
  };
}

{'fraction_non_alphanumeric': 0.12712275594371664, 'fraction_numerical': 0.022319262493934983, 'mean_word_length': 3.623318385650224, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 2, 'lorem ipsum': 0, 'www.': 2, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '416941', 'n_tokens_mistral': 806, 'n_tokens_neox': 764, 'n_words': 141}